
Historia

x 350 a.n.e.: Grecia clásica (Aristóteles,...)

x 1930: Edad de oro de la lógica (Gödel)

x 1960: Demostración automática de teoremas

x 1965: Resolución y unificación (Robinson)

x 1969: QA3, obtención de respuesta (Green)

x 1972: Implementación de Prolog (Colmerauer)

x 1974: Programación lógica (Kowalski)

x 1977: Prolog de Edimburgo (Warren)

x 1981: Proyecto japonés de Quinta Generación

x 1986: Programación lógica con restricciones

x 1995: Estándar ISO de Prolog

Un ejemplo simple: divisibilidad

x Problema: Escribir un programa para declarar
que 2 divide a 6 y utilizarlo para responder a
las siguientes cuestiones:

u ¿2 divide a 6?.

u ¿3 divide a 12?.

u ¿Cuáles son los múltiplos de 2?.

u ¿Cuáles son los divisores de 6?.

u ¿Cuáles son los elementos X e Y tales que X divide a

Y?.

x Programa: divisibilidad-1.pl

divide(2,6).

x Sesión

?- divide(2,6).
Yes
?- divide(3,12).
No
?- divide(2,X).
X = 6
Yes
?- divide(X,Y).
X=2
Y=6
Yes

Ampliación del programa

x Problema: Ampliar el programa anterior,
añadiéndole que 2 divide a 12 y que 3 divide
a 6 y a 12 y utilizarlo para responder a las si-
guientes cuestiones:

u ¿Cuáles son los elementos X e Y tales que X divide a

Y?

u ¿Cuáles son los múltiplos de 2 y de 3?

x Programa: divisibilidad-2.pl

divide(2,6).
divide(2,12).
divide(3,6).
divide(3,12).

x Sesión

?- divide(X,Y).
X = 2 Y = 6 ;
X = 2 Y = 12 ;
X = 3 Y = 6 ;
X = 3 Y = 12 ;
No
?- divide(2,X), divide(3,X).
X = 6 ;
X = 12 ;
No

Reglas

x Problema: Ampliar el programa anterior
añadiéndole que los números divisibles por 2
y por 3 son divisibles por 6 y utilizarlo para
responder a las siguientes cuestiones:

u ¿Cuáles son los múltiplos de 6?

u ¿Cuáles son los elementos X e Y tales que X divide a

Y?

x Programa: divisibilidad-3.pl

divide(2,6).
divide(2,12).
divide(3,6).
divide(3,12).
divide(6,X) :-

divide(2,X),
divide(3,X).

x Interpretación de cláusulas

u Cláusula:

divide(6,X) :- divide(2,X), divide(3,X).

u Fórmula:

(∀X)[divide(2, X) ∧ divide(3, X) → divide(6, X)]

u Interpretación declarativa

u Interpretación procedimental

Reglas

x Sesión

?- divide(6,X).
X = 6 ;
X = 12 ;
No
?- divide(X,Y).
X = 2 Y = 6 ;
X = 2 Y = 12 ;
X = 3 Y = 6 ;
X = 3 Y = 12 ;
X = 6 Y = 6 ;
X = 6 Y = 12 ;
No

Resolución en lógica proposicional

x Programa: leche.pl

es_leche :-
parece_leche,
lo_da_la_vaca.

parece_leche :-
es_blanco,
hay_una_vaca_en_la_etiqueta.

lo_da_la_vaca.
es_blanco.
hay_una_vaca_en_la_etiqueta.

x Sesión
?- es_leche.
yes

x Traza

(1) 0 CALL es_leche?
(2) 1 CALL parece_leche?
(3) 2 CALL es_blanco?
(3) 2 EXIT es_blanco
(4) 2 CALL hay_una_vaca_en_la_etiqueta?
(4) 2 EXIT hay_una_vaca_en_la_etiqueta
(2) 1 EXIT parece_leche
(5) 1 CALL lo_da_la_vaca?
(5) 1 EXIT lo_da_la_vaca
(1) 0 EXIT es_leche

Demostración SLD

¤

:- es blanco,
hay una vaca en la etiqueta.
lo da la vaca.

:- hay una vaca en la etiqueta,
lo da la vaca.

:- lo da la vaca.

:- parece leche, lo da la vaca.

:- hay una vaca en la etiqueta.

:- es leche. es leche :- parece leche, lo da la vaca.

parece leche :-
es blanco,
hay una vaca en la etiqueta.

es blanco.

lo da la vaca.

?

?

?

³³³³³³³³³³³³³³³³³³)

»»»»»»»»»»»»»»»»9

»»»»»»»»»»»»»»»»»»9

?

»»»»»»»»»»»»»»»»»»9

³³³³³³³³³³³³³³³³³³)?

x SLD:
u S: regla de Selección

u L: resolución Lineal

u D: cláusulas Definidas

Traza

x Problema: Utilizar el programa anterior para
calcular los divisores de 6 con el dispositivo
trace y construir el árbol de deducción.

?- trace(divide).
divide/2: call redo exit fail

Yes
?- divide(6,X).
T Call: ( 7) divide(6, _G260)
T Call: ( 8) divide(2, _G260)
T Exit: ( 8) divide(2, 6)
T Call: ( 8) divide(3, 6)
T Exit: ( 8) divide(3, 6)
T Exit: ( 7) divide(6, 6)
X = 6 ;
T Redo: ( 8) divide(3, 6)
T Fail: ( 8) divide(3, 6)
T Redo: ( 8) divide(2, _G260)
T Exit: ( 8) divide(2, 12)
T Call: ( 8) divide(3, 12)
T Exit: ( 8) divide(3, 12)
T Exit: ( 7) divide(6, 12)
X = 12 ;
No

Traza

x Arbol de resolución SLD

[1] divide(2,6).
[2] divide(2,12).
[3] divide(3,6).
[4] divide(3,12).
[5] divide(6,X) :-

divide(2,X),
divide(3,X).

?-divide_a(6,X).

?- divide_a(2,X), divide_a(3,X).

?-divide_a(3,6). ?-divide_a(3,12).

[5]

[1] [2]

[3] [4]

{X/6} {X/12}

Reglas recursivas: naturales.pl

x Problema: Los números naturales se forman a
partir del cero y la función sucesor. De forma
más precisa:
* El 0 es un número natural
* Si n es un número natural, s(n) también lo es
Escribir un programa para decidir si una ex-
presión es un número natural y utilizarlo para
responder a las siguientes cuestiones:

u ¿Es s(s(0)) un número natural?

u ¿Es 2 un número natural?

u ¿Cuáles son los números naturales?

x Programa: naturales.pl

nat(0).
nat(s(X)) :-

nat(X).

Reglas recursivas: naturales.pl

x Sesión

?- nat(s(s(0))).
Yes
?- nat(dos).
No
?- nat(X).
X = 0 ;
X = s(0) ;
X = s(s(0)) ;
X = s(s(s(0))) ;
X = s(s(s(s(0))))
Yes

Reglas recursivas: naturales.pl

x Arbol de resolución SLD

(I) nat(0).
(II) nat(s(X)):-nat(X).

θ2 = {N0 → s(X1)}θ1 = {N0 → 0}

θ21 = {X1 → 0} θ22 = {X1 → s(X2)}

θ221 = {X2 → 0} θ222 = {X2 → s(X3)}

(I)
(II)

(I) (II)

(I) (II)

:-nat(N0).

:-nat(X1).

:-nat(X2).¤
RC2 = {N0 → s(0)}

¤ . . .

RC3 = {N0 → s(s(0))}

¤
RC1 = {N0 → 0}

½

½
½

½
½

½
½

½½=

Z

Z

Z

Z

Reglas recursivas: suma.pl

x Problema: Definir el predicado suma(X,Y,X)

de forma que si X e Y son dos números
naturales con la representación del programa
naturales.pl, entonces Z es el resultado de
sumar X e Y. Por ejemplo,

suma(s(0),s(s(0)),X) => X=s(s(s(0)))

Utilizarlo para responder a las siguientes cues-
tiones:

u ¿Cuál es la suma de s(0) y s(s(0))?

u ¿Cuál es la resta de s(s(s(0))) y s(0)

u ¿Cuáles son las soluciones de la ecuación X + Y =

s(s(0))?

x Programa: suma.pl

suma(0,X,X).
suma(s(X),Y,s(Z)) :- suma(X,Y,Z).

Reglas recursivas: suma.pl

x Sesión

?- suma(s(0),s(s(0)),X).
X = s(s(s(0)))
Yes
?- suma(X,s(0),s(s(s(0)))).
X = s(s(0))
Yes
?- suma(X,Y,s(s(0))).
X = 0
Y = s(s(0)) ;
X = s(0)
Y = s(0) ;
X = s(s(0))
Y = 0 ;
No

Representación de listas

x Śımbolos:
u Una constante: []

u Un śımbolo de función binario: .

x Ejemplos de listas (notación de términos)

u []

u .(a,[])

u .(1,.(2,.(3,.(4,[]))))

u .([],[])

x Ejemplos de listas (notación reducida)

u []

u [a]

u [1,2,3,4]

u [[]]

Representación de listas

x El predicado display

?- display([]).
[]
Yes
?- display([a]).
.(a, [])
Yes
?- display([1,2,3,4]).
.(1, .(2, .(3, .(4, []))))
Yes
?- display([[]]).
.([], [])
Yes

x El predicado de unificación: =

?- X = .(a,.(1,[])).
X = [a, 1]
Yes
?- .(X,Y) = [1].
X = 1
Y = []
Yes
?- .(X,Y) = [a,b,c].
X = a
Y = [b, c]
Yes
?- .(X,.(2,.(Z,[]))) = [1,Y,3].
X = 1
Z = 3
Y = 2
Yes

Unificación con listas

x ¿Son unificables las siguientes listas?

u [X|Y] y [1,2,3]

?- [X|Y] = [1,2,3].
X = 1
Y = [2, 3]
Yes

u [X,Y|Z] y [1,2,3]

?- [X,Y|Z] = [1,2,3].
X = 1
Y = 2
Z = [3]
Yes

u [X,Y] y [1,2,3]

?- [X,Y] = [1,2,3].
No

u [X,Y|Z] y [1,2]

?- [X,Y|Z] = [1,2].
X = 1
Y = 2
Z = []
Yes

Unificación con listas

u [X,a,b|[]] y [[b,L],a|L]

?- [X,a,b|[]] = [[b,L],a|L].
X = [b, [b]]
L = [b]
Yes

u [X,Y,Z] y [Y,Z,X]

?- [X,Y,Z] = [Y,Z,X].
X = _G171
Y = _G171
Z = _G171
Yes

u [X,Y,Z] y [Y,Z|[]]

?- [X,Y,Z] = [Y,Z|[]].
No

u p([X|R],h(X,[a|[R|L]])) y p([a],h(a,[L]))

?- p([X|R],h(X,[a|[R|L]])) = p([a],h(a,[L])).
No

u X y f(X)

?- X = f(X).
Action (h for help) ? a
abort
Execution Aborted

Operaciones con listas

x Problema: Primer elemento y resto de una lista

u primero(L,X) se verifica si X es el primer elemento de

la lista L

u resto(L,X) se verifica si X es el resto de la lista L

u Ejemplo

primero([a,b,c],X) => X=a
resto(([a,b,c],X) => X=[b,c]

u Programa listas-1.pl

primero([X|L],X).
resto([X|L],L).

u Sesión

?- primero([a,b,c],X).
X = a
Yes
?- primero([X,b,c],a).
X = a
Yes
?- primero([X,Y],a).
X = a
Y = _G286
Yes
?- primero(X,a).
X = [a|_G353]
?- resto([a,b,c],L).
L = [b, c]
?- resto([a|L],[b,c]).
L = [b, c]

Operaciones con listas

x Añadir un elemento a una lista
u cons(X,L1,L2) see verifica si L2 es la lista obtenida

añadiéndole X, como primer elemento, a la lista L1

u Ejemplo

cons(a,[b,c],L2) => L2 = [a,b,c]

u Programa cons.pl

cons(X,L1,[X|L1]).

u Sesión

?- cons(a,[b,c],L).
L = [a, b, c]
?- cons(X,L,[a,b,c]).
X = a
L = [b, c] ;

Operaciones con listas

x Concatenación de listas
u conc(L1,L2,L3) se verifica si L3 es la lista obtenida es-

cribiendo los elementos de L2 a continuación de los

elementos de L1.

u Ejemplo

conc([a,b],[c,d].L3) =>, L3 = [a,b,c,d]

u Programa conc.pl

conc([],L,L).
conc([X|L1],L2,[X|L3]) :-

conc(L1,L2,L3).

u Esquema

X

L1

L1

L2

L2 X

L3

L3

Operaciones con listas

u ¿Cuál es el resultado de concatenar las listas [a,b] y

[c,d,e]?

?- conc([a,b],[c,d,e],L).
L = [a, b, c, d, e] ;
No

u ¿Qué lista hay que añadirle al lista [a,b] para obtener

[a,b,c,d]?.

?- conc([a,b],L,[a,b,c,d]).
L = [c, d] ;
No

u ¿Qué dos listas hay que concatenar para obtener

[a,b]?.

?- conc(L1,L2,[a,b]).
L1 = []
L2 = [a, b] ;
L1 = [a]
L2 = [b] ;
L1 = [a, b]
L2 = [] ;
No

Operaciones con listas

u ¿Pertenece b a la lista [a,b,c]?
.

?- conc(L1,[b|L2],[a,b,c]).
L1 = [a]
L2 = [c] ;
No
?- conc(_,[b|_],[a,b,c]).
Yes

u ¿Es [b,c] una sublista de [a,b,c,d]?

?- conc(_,[b,c|_],[a,b,c,d]).
Yes

u ¿Es [b,d] una sublista de [a,b,c,d]?

?- conc(_,[b,d|_],[a,b,c,d]).
No

u ¿Cuál es el último elemento de [b,a,c,d]?.

?- conc(_,[X],[b,a,c,d]).
X = d ;
No

u Predicado predefinido: append(L1,L2,L3)

Aritmética

x Operadores prefijos e infijos

u (a + b) * (5 / c)

?- display((a + b) * (5 / c)).
*(+(a, b), /(5, c))
Yes

u a + b * 5 / c

?- display(a + b * 5 / c).
+(a, /(*(b, 5), c))
Yes

x Precedencia y tipo de operadores predefinidos:

+-------------+------+-------------+
| Precedencia | Tipo | Operadores |
+-------------+------+-------------+
| 500 | yfx | +, - |
| 500 | fx | - |
| 400 | yfx | *, / |
| 200 | xfy | ^ |
+-------------+------+-------------+

u fx, fy: Prefijo

u xfx: Infijo no asociativo

u yfx: Infijo asocia por la izquierda

u xfy: Infijo asocia por la derecha

u xf, yf: Postfijo

Aritmética

x Análisis de expresiones con operadores

?- display(2+3+4).
+(+(2, 3), 4)
Yes

?- display(2+3*4).
+(2, *(3, 4))
Yes

?- display((2+3)*4).
*(+(2, 3), 4)
Yes

?- display(2^3^4).
^(2, ^(3, 4))
Yes

Predicados aritméticos

u Evaluador: is
u Predicados aritméticos:

+-------------+------+---------------------------+
| Precedencia | Tipo | Operadores |
+-------------+------+---------------------------+
| 700 | xfx | <, =<, >, >=, =:=, =\= |
+-------------+------+---------------------------+

u Ejemplos

?- X is 2+3^3.
X = 29
Yes
?- 29 is X+3^3.
[WARNING: Arguments are not sufficiently instantiated]
?- X = 2+3^3.
X = 2+3^3
Yes
?- 2+3^Y = 2+3^3.
Y = 3
Yes
?- 3 =< 5.
Yes
?- 3 > X.
[WARNING: Arguments are not sufficiently instantiated]
?- 2+5 = 10-3.
No
?- 2+5 =:= 10-3.
Yes
?- 2+5 =\= 10-3.
No
?- 2+5 =\= 10^3.
Yes

Máximo

x Máximo de dos números
u maximo(X,Y,Z) se verifica si Z es el máximo de los

números X e Y.

u Ejemplo

maximo(3,5,Z) => Z=5

u Programa: maximo.pl

maximo(X,Y,X) :-
X >= Y.

maximo(X,Y,Y) :-
X < Y.

u Sesión

?- maximo(2,3,X).
X = 3 ;
No
?- maximo(3,2,X).
X = 3 ;
No

Factorial

x Factorial de un número
u factorial(X,Y) se verifica si Y es el factorial de X

u Programa: factorial.pl

factorial(1,1).
factorial(X,Y) :-

X > 1,
X1 is X - 1,
factorial(X1,Y1),
Y is X * Y1.

u Sesión

?- factorial(4,Y).
Y = 24
Yes

u Cálculo de factorial(4,Y)

X = 4
X1 is X-1 => X1 = 3
factorial(X1,Y1) => Y1 = 6
Y is X*Y1 => Y = 24

Fibonacci

x Sucesión de Fibonacci
u La sucesión de Fibonacci es

0, 1, 1, 2, 3, 5, 8, ...

y está definida por

f(0) = 0

f(1) = 1

f(n) = f(n-1)+f(n-2), si n > 1

u fibonacci(N,X) se verifica si X es el N–ésimo término

de la sucesión de Fibonacci.

u Programa: fibonacci.pl

fibonacci(0,0).
fibonacci(1,1).
fibonacci(N,X) :-

N > 1,
N1 is N-1,
fibonacci(N1,X1),
N2 is N-2,
fibonacci(N2,X2),
X is X1+X2.

u Sesión

?- fibonacci(6,X).
X = 8
Yes

Longitud

x Longitud de una lista

u longitud(L,N) se verifica si N es la longitud de la lista

L

u Ejemplos

longitud([],N) => N = 0
longitud([a,b,c],N) => N = 3
longitud([a,[b,c]],N) => N = 2

u Programa: longitud.pl

longitud([],0).
longitud([_X|L],N) :-

N is M + 1.

u Sesión

?- longitud([a,b,c],N).
N = 3
Yes

u Predicado predefinido: length(L,N)

Máximo de una lista

x Máximo de una lista
u max list(L,N) se verifica si N es el máximo de los ele-

mentos de la lista de números L

u Sesión

?- max_list([1,3,9,5],X).
X = 9
Yes

u Programa: max list.pl

max_list([X],X).
max_list([X,Y|L],Z) :-

max_list([Y|L],U),
maximo(X,U,Z).

maximo(X,Y,X) :-
X >= Y.

maximo(X,Y,Y) :-
X < Y.

Entre

x Intervalo entero
u entre(N1,N2,X) que se verifica si X es mayor o igual que

N1 y menor o igual que N2.

u Sesión

?- entre(2,5,X).
X = 2 ;
X = 3 ;
X = 4 ;
X = 5 ;
No
?- entre(2,1,X).
No

u Programa: entre.pl

entre(N1,N2,N1) :-
N1 =< N2.

entre(N1,N2,X) :-
N1 < N2,
N3 is N1 + 1,
entre(N3,N2,X).

u Predicado predefinido: between(N1,N2,X)

Operadores

x Operadores definidos por el ususario

u pedro come pan ==> come(pedro,pan)

u juan tiene discos ==> tiene(juan,discos)

x Definición de operadores (directiva op)

:- op(600,xfx,come).

x Precedencia: Entre 1 y 1200

Operadores

x Estructura de arbol

Juan y Pedro comen pan
comen

y pan

juan pedro

u Sin operadores:

comen(y(juan,pedro),pan).

u Con operadores:

:-op(800,xfx,comen).
:-op(400,xfx,y).
juan y pedro comen pan.

?- display(juan y pedro comen pan).
comen(y(juan, pedro), pan)
Yes

?- Quienes comen pan.
Quienes = juan y pedro ;
No
?- Alguien y pedro comen pan.
Alguien = juan ;
No
?- juan y pedro comen Algo.
Algo = pan ;
No

Operadores

x Precedencia entre argumentos

u La precedencia de un término simple es cero

u La precedencia de un término compuesto es la del

śımbolo de función principal y cero si no está pre-

definido

x x e y

u x representa un argumento cuya precedencia es estric-

tamente menor que la del operador

u y representa un argumento cuya precedencia es menor

o igual que la del operador

x Ejemplo:

El libro de ciencias de Juan es rojo
es

rojode

juande

el_libro ciencias

Operadores

u Programa

:-op(800,xfx,es).
:-op(400,yfx,de).
el_libro de ciencias de juan es rojo.

?- display(el_libro de ciencias de juan es rojo).
es(de(de(el_libro, ciencias), juan), rojo)
Yes

?- display(X es rojo).
es(_G177, rojo)
X = _G177
Yes

?- X es rojo.
X = el_libro de ciencias de juan
Yes

Operadores

?- display(X de Y es rojo).
es(de(X, Y), rojo)
Yes

?- display(el_libro de ciencias de juan es rojo).
es(de(de(el_libro, ciencias), juan), rojo)
Yes

?- X de Y es rojo.
X = el_libro de ciencias
Y = juan
Yes

?- display(el_libro de X es rojo).
es(de(el_libro, X), rojo)
Yes

?- el_libro de X es rojo.
No

